
FICHA METODOLÓGICA 

NOMBRE DEL INDICADOR Índice de Pobreza Multidimensional 

DEFINICIÓN 

Es un índice que identifica el conjunto de privaciones 

de derechos a nivel de los hogares en 4 dimensiones y 

refleja la proporción de personas pobres 

multidimensionales y el porcentaje promedio de 

privaciones que cada persona sufre de manera 

simultánea.  

El índice está acotado entre 0 y 1, donde 1 significa que 

todos los hogares son pobres multidimensionalmente 

en todos los indicadores y 0 en caso que ningún hogar 

sea pobre multidimensional. 

FÓRMULA DE CÁLCULO 

𝐼𝑃𝑀 = 𝑇𝑃𝑀 × 𝐼𝑛𝑡𝑒𝑛𝑠𝑖𝑑𝑎𝑑 
 
Dónde: 
IPM: Índice de pobreza multidimensional 
TPM: Tasa de la pobreza multidimensional  
Intensidad: Intensidad de la pobreza  

DEFINICIÓN DE LAS VARIABLES RELACIONADAS 

Tasa de Pobreza Extrema Multidimensional (TPEM).- Porcentaje de personas que viven en 

hogares que tienen privaciones en al menos la mitad (K >= 50%) de los indicadores 

ponderados. 

Tasa de Pobreza Multidimensional (TPM).- Porcentaje de personas que viven en hogares 

que tienen privaciones en una tercera parte o más (K >= 33.3%) de los indicadores 

ponderados. 

Intensidad de la pobreza (A): Corresponde al porcentaje promedio de privaciones de los 

pobres multidimensionales. 

 

Inasistencia a educación básica y bachillerato.- Se consideran privadas en el derecho a la 

educación los niños y niñas entre 5 a 14 años que no asisten a un centro de educación básica 

y también los jóvenes entre 15 a 17 años que no asisten al bachillerato. 

No acceso a educación superior por razones económicas.- Se categorizan como privadas al 

derecho a la educación a los jóvenes entre 18 y 29 años que habiendo terminado el 

bachillerato, no pueden acceder a un centro de educación superior de tercer nivel debido a 

la falta de recursos económicos. 

 

 

Logro educativo incompleto.- Se consideran privados en el derecho a la educación las 

personas entre 18 a 64 años, que no hayan terminado la educación básica, es decir, que 

tengan menos de 10 años de escolaridad y que no asistan a un centro de educación formal. 

 

No contribución al sistema de pensiones.- Se consideran privadas las personas ocupadas de 

15 años o más, que no aportan a ningún tipo de seguridad social; excluyendo de la privación 

a personas ocupadas de 65 años y más, que no aportan, pero reciben pensión por jubilación. 

Para personas en condición de desempleo o económicamente inactivas, de 65 años o más, 

se las considera en privación si no reciben pensión por jubilación, Bono de Desarrollo 

Humano o Bono Joaquín Gallegos Lara. 

 

Empleo infantil y adolescente.- Todo los niños y niñas entre 5 a 14 años que estén ocupados 

en la semana de referencia se identifican como privados al considerarse prohibido el trabajo 

infantil. Para los adolescentes entre 15 a 17 años, se los considera privados al derecho al 

trabajo si, estando ocupados en la semana de referencia cumplen una de las siguientes 

condiciones: recibieron una remuneración inferior al Salario Básico Unificado, no asisten a 

clases o trabajaron más de 30 horas. 

 

Desempleo o empleo inadecuado.- Se consideran privadas en su derecho al trabajo a las 

personas de 18 años o más, que en el período de referencia, estuvieron desocupadas. 

Adicionalmente, se consideran privadas las personas ocupadas que tienen un empleo 

inadecuado. 

 

Sin servicio de agua por red pública.- Se identifica como privados a los miembros de las 

viviendas que obtienen el agua por un medio distinto al de la red pública. 

 

Pobreza extrema por ingresos.- Se consideran privadas a las personas cuyo ingreso per cápita 

familiar es inferior al de la línea de pobreza extrema.  

 

Hacinamiento.- Se encuentran en condición de hacinamiento, los miembros de viviendas 

que tienen más de tres personas por dormitorio exclusivo para dormir. 

Déficit habitacional.- se consideran en déficit habitacional las personas cuya vivienda, debido 

a los materiales o estado de sus paredes, piso y techo, son consideradas en déficit 

cualitativo1 o cuantitativo2. 

 

 

 

Sin saneamiento de excretas.- Se catalogan como privadas en saneamiento de excretas a las 

personas del área urbana cuya vivienda no cuenta con servicio higiénico conectado a 

alcantarillado. En el área rural, las personas privadas son aquellas cuya vivienda no cuenta 

con alcantarillado o pozo séptico. 

 

Sin servicio de recolección de basura.- Las personas que habitan en viviendas que no tienen 

acceso al servicio municipal de recolección de basura, se clasifican como privadas en este 

indicador. 

Nota: Las variables se construyen en base a la ENEMDU.  

METODOLOGÍA DE CÁLCULO 

 
A continuación se presenta un extracto de la Metodología del Índice de Pobreza 
Multidimensional disponible en el segundo volumen de la Revista de Estadística y 
Metodologías. Citar como: (INEC (2016), Revista de Estadística y Metodologías. Ecuador) 
 
1. Selección de las dimensiones:  
El índice agrupa cuatro dimensiones: i) educación, ii) trabajo y seguridad social, iii) salud, 

agua y alimentación, y iv) hábitat, vivienda y ambiente sano.  

2. Selección de indicadores: 
Dentro de cada dimensión se construyen indicadores. En total existen 12 indicadores 

distribuidos de la siguiente manera.  

Tabla 1: Indicadores del Índice de Pobreza Multidimensional Ecuador 

Dimensión Indicador 

Educación 

1. Inasistencia a educación básica y bachillerato 

2. No acceso a educación superior por razones 
económicas 

3. Logro educativo incompleto 

Trabajo y  

Seguridad social 

4. No contribución al sistema de pensiones 

5. Empleo infantil y adolescente 

6. Desempleo o empleo inadecuado  

7. Sin servicio de agua por red pública 

                                                           
1 Viviendas que presentan carencias habitacionales en los atributos referentes a la estructura y requieren 
mejoramiento. (Ficha Metodológica: INEC) 
2 Unidad habitacional que necesita ser reemplazada por una nueva. (Ficha Metodológica: Déficit cuantitativo de 
vivienda, INEC) 

http://app.sni.gob.ec/sni-link/sni/Portal%20SNI%202014/FICHAS%20INDICADORES%20PNBV_2013-2017/Obj%203/Indicadores%20Apoyo/A3.16.%20D%C3%A9ficit%20habitacional%20cualitativo.pdf
http://portal.sni.gob.ec/documents/10156/9b61a3c6-89c2-4b44-a3c2-0392d82d8c0f

Salud, Agua y 

Alimentación 
8. Pobreza extrema por ingresos 

Hábitat, Vivienda 

y  

Ambiente sano 

9. Hacinamiento 

10. Déficit habitacional 

11. Sin saneamiento de excretas  

12. Sin servicio de recolección de basura 

 
Por cada indicador se determinan privaciones.  Si un individuo está afectado en su derecho 
según los indicadores seleccionados, se le identifica con el valor de 1, cero en caso contrario.

 

 
 
3. Agregación de privaciones:  

Si en el hogar al menos uno de sus miembros está privado los demás miembros del 
hogar heredan dicha característica.  
 

4. Estructura de la ponderación 
Cada dimensión recibe un peso del 25% y dentro de cada dimensión el peso se 
distribuye proporcionalmente. Luego cada indicador se multiplica por su peso 
respectivo teniendo un recuento ponderado de privaciones. 
 

5. Línea de pobreza multidimensional (k) 
Se define a una persona como pobre cuando su hogar tiene privaciones en una tercera 

parte o más (k>=33.3%) de los indicadores ponderados. Se define a una persona como 

pobre extrema multidimensional cuando su hogar tiene privaciones en la mitad o más 

de los indicadores ponderados (k>=50%). 

 
6. Índice de Pobreza Multidimensional -IPM 

En índice es el resultado de ajustar la Tasa de Pobreza Multidimensional por la 

Intensidad de la pobreza. El índice está acotado entre 0 y 1, donde 1 significa que todos 

los hogares son pobres multidimensionalmente en todos los indicadores y 0 en caso que 

ningún hogar sea pobre multidimensional. 

UNIDAD DE MEDIDA O EXPRESIÓN DEL 

INDICADOR 

El índice de pobreza multidimensional es 

adimensional está acotado entre cero y 

uno. 

INTERPRETACIÓN DEL INDICADOR 

El índice refleja el porcentaje de pobres 

multidimensionales ajustado por las 

privaciones promedio que sufren los 

pobres multidimensionales. Dicho de otra 

manera, el índice mide qué tan pobres son 

los pobres. 

FUENTE DE DATOS 
Encuesta Nacional de Empleo, Desempleo 

y Subempleo-ENEMDU 

PERIODICIDAD DEL INDICADOR  Anual 

DISPONIBILIDAD DE LOS DATOS A partir del 2009.  

NIVEL DE 

DESAGREGACIÓN 

GEOGRÁFICO 
Área de residencia, región natural, zonas 

de planificación. 

GENERAL 
Sexo, etnia, nivel de escolaridad, categoría 

de ocupación del jefe del hogar.  

OTROS ÁMBITOS  

INFORMACIÓN GEO – REFERENCIADA No aplica 

RELACIÓN CON INSTRUMENTOS DE 

PLANIFICACIÓN NACIONAL E INTERNACIONAL 

Plan Nacional del Buen Vivir  2013-2017 

Objetivo 2: “superar las definiciones 

clásicas de pobreza, medidas por el 

ingreso, el consumo (monetario) […] y 

profundizar una concepción 

multidimensional que dé cuenta del 

ejercicio real e igualitario de derechos”. 

Constitución del Ecuador Asamblea 

Nacional  

Artículos del 12 a 34.  

REFERENCIAS BIBLIOGRÁFICAS DE LA 

CONSTRUCCIÓN DEL INDICADOR 

Alkire, S., & Foster, J. (2011). Counting and 

multidimensional poverty measurement. 

Journal of public economics, 95(7), 476-

487. 

FECHA DE ELABORACIÓN DE LA FICHA 

METODOLÓGICA 
29/01/2016 

FECHA DE LA ÚLTIMA ACTUALIZACIÓN DE LA 

FICHA 
02/02/2016 

CLASIFICADOR SECTORIAL 

Dominio 3: 

Estadísticas 

ambientales y 

varios dominios 

3.3 Estadísticas e 
indicadores 
Multidominio 

3.3.1 Condiciones de 

vida, pobreza y 

problemas sociales 

ELABORADO POR 

Instituto Nacional de Estadística y Censos 

(INEC) 

Dirección de Innovación en Métricas y 

Metodologías (DINME) 

Fausto Jácome  
fausto_jacome@inec.gob.ec  
Roberto Castillo A. 
 roberto_castillo@inec.gob.ec 
José C. Andrade 
jose_andrade@inec.gob.ec  
Melody Serrano 
melody_serrano@inec.gob.ec 

ANEXOS 

ANEXO 1 : Algoritmo de cálculo del Indicador 

SINTAXIS 

*=================================================================================* 
* TÍTULO DE LA SINTAXIS:       
* Cálculo de Índice de Pobreza Multidimensional – IPM 
* OPERACIÓN ESTADÍSTICA:         
* Encuesta Nacional de Empleo, Desempleo y Subempleo (ENEMDU)   
* UNIDAD TÉCNICA RESPONSABLE:        
* Dirección de Innovación en Métricas y Metodologías (DINME)        
* ENTIDAD EJECUTORA:          
* Instituto Nacional de Estadística y Censos (INEC) 
*=================================================================================* 
* Fecha de elaboración:        12 de Noviembre del 2015         
* Fecha última modificación:   26 de Enero del 2016         
*=================================================================================* 
* Elaborado por:          
* Fausto Jácome, Roberto Castillo, José Andrade, Cristina Restrepo y Melody Serrano  
*=================================================================================* 
* Revisado por:           

* Equipo Técnico de la Comisión Especial de Estadística de Pobreza     
* Oxford Poverty & Human Development Iniciative       
*=================================================================================* 
* Aprobado por:           
* Roberto Castillo A.          
* Director de Innovación en Métricas y Metodologías (DINME)        
* Instituto Nacional de Estadística y Censos (INEC)       
* roberto_castillo@inec.gob.ec         
*=================================================================================* 
 
*=================================================================================* 
* Instrucciones: 
* 1. Descargar las bases de la página: 
*    www.ecuadorencifras.gob.ec/banco-de-informacion/ 
*    Pestaña  -Investigación: Todas 
*       -Tipo de archivo: Bases 
*       -Archivo:  
*    Producto - Encuesta Nacional de Empleo Desempleo Subempleo 
*   Tipo de encuesta - 15 años 
*    Período - Trimestral 
*    Archivo - bdd_enemdu_15anios_12_(seleccionar año) 
* 2. Transformar las bases de personas (*12_enemdubdd_15anios) y vivienda  
*    (*12_enemdubdd_viviendahogar) de SPSS (*.sav) a formato STATA (*.dta). 
*  En caso de tener problemas, revisar el manual de la sintaxis del IPM. 
* 3. Guardar en un directorio y copiar ese directorio en "cd". 
* 4. Correr la programación en stata. 
*=================================================================================* 
clear all 
set more off 
* Cambiar Directorio 
cd "C:\..." 
* Atención.- No modificar el nombre de las bases descargadas 
* Nombre de la base de PERSONAS descargada 
loc bdd_per "*12_enemdubdd_15anios" //reemplazar asterisco por el año 
* Nombre de la base de VIVIENDA descargada 
loc bdd_viv "*12_enemdubdd_viviendahogar" //reemplazar asterisco por el año  
 
*=================================================================================* 
* PREPARACIÓN DE LA BASE DE TRABAJO 
*=================================================================================* 
* Unir bases de personas y hogares 
loc bdd=substr("`bdd_per'",1,4) 
loc n1=1 
foreach base in `bdd_per' `bdd_viv' { 
use `base', clear 
cap drop id_upm  
cap drop id_hogar  
 
if `n1'==1 loc varid "ciudad zona sector panelm vivienda hogar p01" 
else    loc varid "ciudad zona sector panelm vivienda hogar" 
 
 foreach var of loc varid { 
 tempvar largo 
 tostring `var', replace force 

 gen `largo'=length(`var') 
 sum `largo' 
 replace `var'=substr(string(10^(r(max)-`largo')),2,.)+`var' if r(max)>`largo' 
 drop `largo' 
 } 
* Creación de identificadores únicos 
egen id_hogar = concat(ciudad zona sector panelm vivienda hogar), punct("-") 
label var id_hogar "Identificador hogar" 
if `n1'==1 { 
egen id_per=    concat(ciudad zona sector panelm vivienda hogar p01), punct("-") 
label var id_per "Identificador persona" 
} 
destring `varid', replace 
tempfile bdd`n1' 
save `bdd`n1++'', replace 
} 
use `bdd1', clear 
merge m:1 id_hogar using `bdd2', keep(1 3) nogen 
 
*=================================================================================* 
* A. CREACIÓN DE DIMENSIONES E INDICADORES 
*=================================================================================* 
 
*=================================================================================* 
* DIMENSION 1: EDUCACIÓN 
*=================================================================================* 
 
*=================================================================================* 
* 1. Inasistencia a educación básica y bachillerato 
*=================================================================================* 
/*  
Población de referencia: Niños (5-14 años) y Adolescentes (15-17 años) 
Privación: 
(5-14 años): No asiste a centro de educación básica. 
(15-17 años): Completó educación básica y no asiste al bachillerato. 
*/ 
 
* a) Asistencia a educación básica 
# delimit; 
gen     asist_basica=1 if inrange(p03,5,14) & (p07==1) &  
(  inlist(p10a,1,3) 
| (p10a==4 & inrange(p10b,0,6))  
| (p10a==5 & inrange(p10b,0,9))  
| (p10a==6 & inrange(p10b,0,2))); 
# delimit cr 
replace asist_basica=0 if inrange(p03,5,14) & asist_basica!=1 
label var asist_basica "Asistencia a educación básica" 
 
* b) Asistencia a educación bachillerato 
# delimit; 
gen     asist_bach=1 if inrange(p03,15,17) & (p07==1)  &  
( (p10a==5 & p10b==10) 
| (p10a==7 & inrange(p10b,0,2))  
| (p10a==6 & inrange(p10b,3,5))); 

# delimit cr 
replace asist_bach=0 if inrange(p03,15,17) & asist_bach!=1  
label var asist_bach "Asistencia a educación bachillerato" 
 
* c) Cálculo de la privación          
           
gen   dim1_ind1=1 if inrange(p03,5,17) & p10a<8  
replace dim1_ind1=0 if inrange(p03,5,17) & (asist_basica==1 | asist_bach==1 | p10a>=8) 
label var dim1_ind1 "Inasistencia a educación básica y bachillerato" 
 
*=================================================================================* 
* 2. No acceso a educación superior por barreras económicas 
*=================================================================================* 
/*  
Población de referencia: Jóvenes (18-29 años) bachilleres. 
Privación:  
No asiste a un establecimiento de educación superior por falta de recursos económicos. 
*/ 
 
# delimit;          
          
gen     dim1_ind2=0 if inrange(p03,18,29) & (p07>=1 & p07<=2) & 
( (p10a==7 & p10b>=3)  
| (p10a==6 & p10b>=6))  
| inrange(p10a,8,10);  
# delimit cr  
 
# delimit;          
          
replace   dim1_ind2=1 if inrange(p03,18,29) & p07==2 & 
( (p10a==7 & p10b>=3)  
| (p10a==6 & p10b>=6)  
| (inrange(p10a,8,9))) 
& p09==3 ;  
# delimit cr  
replace   dim1_ind2=. if p03<18 | p03>29 
label var dim1_ind2 "Barrera económica al acceso a educación superior" 
 
*=================================================================================* 
* 3. Logro educativo incompleto 
*=================================================================================* 
/* 
Población de referencia: Personas de 18 a 64 años. 
Privación:  
Personas con menos de 10 años de escolaridad que no asisten 
a centros de educación formal. 
*/ 
 
* a) Años promedio de escolaridad 
gen     escol=. 
replace escol=0    if p10a==1 & p10b==0 
replace escol=0    if p10a==2 & p10b==0  
replace escol=2    if p10a==2 & p10b==1 
replace escol=4    if p10a==2 & p10b==2 

replace escol=6    if p10a==2 & p10b==3 
replace escol=7    if p10a==2 & p10b==4 
replace escol=1    if p10a==3 
replace escol=1+ p10b if p10a==4 
replace escol=   p10b if p10a==5 
replace escol=7+ p10b if p10a==6 
replace escol=10+p10b if p10a==7 
replace escol=13+p10b if p10a==8 
replace escol=13+p10b if p10a==9 
replace escol=18+p10b if p10a==10 
replace escol=0    if escol==. 
 
* b) Cálculo de la privación 
gen     dim1_ind3=0 if inrange(p03,18,64) & ((escol<10 & p07==1) | (escol>=10)) 
replace dim1_ind3=1 if inrange(p03,18,64) &   escol<10 & p07==2 
label var dim1_ind3 "Logro educativo incompleto" 
 
*=================================================================================* 
* DIMENSION 2: TRABAJO Y SEGURIDAD SOCIAL 
*=================================================================================* 
 
*=================================================================================* 
* 1. Empleo infantil y adolescente 
*=================================================================================* 
/*  
Población de referencia: Niños (5-14 años) y Adolescentes (15-17 años) 
Privación:  
Niños (5-14 años): Ocupado en la semana de referencia 
Adolescentes (15-17 años): Ocupado en la semana de referencia y cumple una  
de las siguientes condiciones: 
i)  Recibe una remuneración inferior al Salario Básico Unificado 
ii) No asisten a clases 
iii) Trabajan más de 30 horas. 
*/ 
 
gen     horas =0 if empleo ==1 
replace horas =p24 if pea ==1 & p20 ==1 
replace horas =p24 if pea ==1 & p20 ==2 & p21 <=11 
recode p51* (999 = .) 
egen hh = rowtotal (p51*), missing 
replace hh  = . if hh <0 
replace horas = hh if pea ==1 & p20 ==2 & p21 ==12 & p22==1 
 
gen     dim2_ind1=(p20==1 | inrange(p21,1,11) | p22==1) & inrange(p03, 5,14) 
replace dim2_ind1=1 if inrange(condactn,2,6) & inrange(p03,15,17) 
replace dim2_ind1=1 if         condactn ==1  & inrange(p03,15,17) & (p07==2 | horas>30) 
replace dim2_ind1=. if p03>=18        
      
label var dim2_ind1 "Empleo infantil y adolescente" 
 
*=================================================================================* 
* 2. Desempleo o empleo inadecuado 
*=================================================================================* 
/*  

Población de referencia: Población de 18 años y más. 
Privación: 
Personas en condición de desempleo o empleo inadecuado. 
*/ 
 
gen     dim2_ind2= inrange(condactn,2,8) & inrange(p03,18,98) 
replace dim2_ind2=. if p03<18 | p03==99 | /// 
        (p20==. & p21==. & p22==. & p32==. & p34==. & p35==.) 
label var dim2_ind2 "Desempleo o empleo inadecuado" 
 
*=================================================================================* 
* 3. No contribución al sistema de pensiones 
*=================================================================================* 
/*  
Población de referencia: Personas de 15 años y más. 
Privación:  
Ocupados que no contribuyen al sistema de seguridad social. Se excluye a ocupados 
de 65 años y más, que no aportan, pero reciben pensión por jubilación. 
Desocupados e inactivos mayores de 65 años que: 
No recibe pensión por jubilación, BDH y Bono Joaquín Gallegos Lara (BJGL).  
*/ 
 
if `bdd'==2009 replace p77=2 // El BJGL se crea en 2010 (decreto ejecutivo N°422) 
 
* Personas ocuapadas 
gen     dim2_ind3 = (empleo==1 & inrange(p05a,5,10) & inrange(p05b,5,10)) 
replace dim2_ind3 = 0 if empleo ==1 & p03>=65 & p72a==1 
 
* Personas desocupadas o PEI mayores de 65 años  
replace dim2_ind3 = 1 if (desem ==1 | pei==1) & p03>=65 & p72a==2 
replace dim2_ind3 = 0 if p03>=65 & p72a== 2 & p75==1 
replace dim2_ind3 = 0 if pet ==1 & p77==1 
replace dim2_ind3 = . if pet ==0 | inrange(p03,0,14) 
 
label var dim2_ind3 "No contribución al sistema de pensiones" 
 
*=================================================================================* 
* DIMENSION 3: SALUD, AGUA Y ALIMENTACIÓN 
*=================================================================================* 
 
*=================================================================================* 
* 1. Pobreza extrema por ingresos 
*=================================================================================* 
/*  
Población de referencia: Toda la población. 
Privación:  
Ingreso per cápita familiar es inferior al de la línea de pobreza extrema. 
*/ 
 
* Estimación de la tasa de pobreza extrema 
gen       dim3_ind1 = epobre 
label var dim3_ind1 "Pobreza extrema por ingresos" 
 
*=================================================================================* 



 

12 
 

* 2. Sin servicio de agua por red pública 
*=================================================================================* 
/* 
Población de referencia: Toda la población. 
Privación:  
Viviendas que obtienen el agua por un medio distinto al de la red pública 
*/ 
 
 
gen     dim3_ind2=(vi10!=1)  
replace dim3_ind2=. if vi10 ==. 
label var dim3_ind2 "Sin servicio de agua por red pública" 
 
*=================================================================================* 
* DIMENSION 4: HABITAT, VIVIENDA Y AMBIENTE SANO 
*=================================================================================* 
 
*=================================================================================* 
* 1. Hacinamiento 
*=================================================================================* 
/*  
Población de referencia: Toda la población. 
Privación:  
El número de personas por dormitorio exclusivo para dormir mayor a tres. 
*/ 
 
recode vi07 (0=1) 
egen hsize=count(p01), by(id_hogar) 
gen  dim4_ind1=1 if (hsize/vi07)> 3 
replace dim4_ind1=0 if (hsize/vi07)<=3 
label var dim4_ind1 "Hacinamiento" 
 
*=================================================================================* 
* 2. Déficit habitacional 
*=================================================================================* 
/*  
Población de referencia: Toda la población 
Privación: 
Se consideran en déficit habitacional las personas cuya vivienda, debido a los 
materiales o estado de sus paredes, piso y techo, son consideradas en déficit 
cualitativo  o cuantitativo . 
*/ 
 
* Material y condición del techo 
# delimit ;  
    gen techo=1 if 
 ((vi03a==1 & inrange(vi03b,1,2))  
 |(vi03b==1 & inrange(vi03a,2,4)))  
 ;  
 replace techo=2 if 
 ((vi03a==1 & vi03b==3) 
 | vi03b==2 & inrange(vi03a,2,4)) 
 ;  
 replace techo=3 if 

 ((vi03b==3 & inrange(vi03a,2,4)) 
 |(vi03a==5 | vi03a==6)) 
 ; 
    # delimit cr 
   
* Material y condición de las paredes 
 # delimit ; 
 gen pared=1 if  
 ((vi05a==1 & inrange(vi05b,1,2)) 
 |(vi05a==2 & vi05b==1)) 
 ; 
 replace pared=2 if 
 ((vi05a==1 & vi05b==3) 
 |(vi05a==2 & vi05b==2) 
 |(inrange(vi05a,3,5) & inrange(vi05b,1,2)))  
 ; 
 replace pared=3 if 
  ((vi05b==3 & inrange(vi05a,2,5)) 
 |(vi05a==6 | vi05a==7))      
 ; 
    # delimit cr 
  
* Material y condición del piso 
 # delimit ; 
 gen     piso=1 if 
 ((vi04a<=3 & vi04b==1) 
 |(vi04a<=3 & vi04b==2) 
 |(vi04b==1 & inrange(vi04a,4,5)))  
 ; 
    replace piso=2 if 
 ((vi04a <= 3 & vi04b == 3) 
 |(vi04b==2 & inrange(vi04a,4,5))  
 |(vi04a==6 & vi04b==1)) 
 ; 
 replace piso=3 if 
 ((vi04b==3 & inlist(vi04a,4,5,6)) 
 |(vi04a==6 & vi04b==2) 
 |(vi04a==7 | vi04a==8)) 
 ; 
 # delimit cr 
   
* Tipología de Vivienda 
 # delimit ; 
 gen     tipviv=1 if 
 ((techo==1 & pared==1 & inrange(piso,1,3)) 
 |(techo==1 & pared==2 & piso==1)) 
 ; 
 replace tipviv=2 if 
 ((techo==1 & pared==2 & inrange(piso,2,3)) 
 | techo==1 & pared==3 & inrange(piso,1,2) 
    | techo==2 & pared==1 & inrange(piso,1,2) 
 | techo==2 & pared==2 & inrange(piso,1,2) 
 | techo==3 & pared==1 & inrange(piso,1,2) 
 |(techo==3 & pared==2 & piso==1)) 

 ; 
 replace tipviv=3 if  
 ((techo==1 & pared==3 & piso==3) 
 |(techo==2 & pared==1 & piso==3) 
 |(techo==2 & pared==2 & piso==3) 
 | techo==2 & pared==3 & inrange(piso,1,3) 
 |(techo==3 & pared==1 & piso==3) 
 | techo==3 & pared==2 & inrange(piso,1,3) 
 | techo==3 & pared==3 & inrange(piso,1,3)) 
 ; 
 # delimit cr 
   
label define tipviv 1 "Aceptables" 2 "Recuperables" 3 "Irrecuperables" 
label values tipviv tipviv 
 
* Personas que habitan en viviendas con déficit habitacional 
    gen dim4_ind2=1 if (tipviv==3 | tipviv==2) 
replace dim4_ind2=0 if  tipviv==1 
label var dim4_ind2 "Déficit habitacional" 
   
*=================================================================================* 
* 3. Sin saneamiento de excretas 
*=================================================================================* 
/* 
Población de referencia: Toda la población. 
Privación:  
Área urbana: Vivienda con inodoro conectado a pozo séptico, pozo ciego, letrina  
       o no tiene ningún servicio higiénico. 
Área rural: Vivienda con inodoro conectado a pozo ciego, letrina o no tiene  
   ningún servicio higénico. 
*/ 
 
gen  dim4_ind3=(area==1 & inrange(vi09,2,5)) | (area==2 & inrange(vi09,3,5)) 
replace dim4_ind3=. if vi09==. | area==. 
label var dim4_ind3 "Sin saneamiento de excretas" 
 
*=================================================================================* 
* 4. Sin servicio de recolección de basura 
*=================================================================================* 
/*  
Población de referencia: Toda la población. 
Privación: El hogar no cuenta con servicio municipal para eliminación de basura. 
*/ 
 
gen  dim4_ind4=inlist(vi13,1,3,4,5) 
replace dim4_ind4=. if vi13==. 
label var dim4_ind4 "Sin servicio de recolección de basura" 
 
*=================================================================================* 
* B. IDENTIFICACIÓN Y AGREGACIÓN DE PRIVACIONES       
*=================================================================================* 
 
foreach var of varlist dim* { 
rename `var' `var'_p 

bys id_hogar: egen `var'_h=max(`var'_p) 
local a: variable label `var'_p 
label variable `var'_p "`a' -grupo específico" 
label variable `var'_h "`a' -nivel de hogar" 
order `var'_p `var'_h, last 
} 
recode dim*_h (.=0) 
 
tempfile bdd3 bddf  
save `bdd3', replace 
 
/* Tratamiento de valores extremos: Se excluyen hogares de la muestra si no 
   pueden ser evaluados - Missing data*/ 
 
*1.Educación* 
egen miss_dim1_ind1=rowmiss(p07 p10a) if inrange(p03,5,17) 
egen miss_dim1_ind2=rowmiss(p07 p10a) if inrange(p03,18,29) 
egen miss_dim1_ind3=rowmiss(p07 escol) if inrange(p03,18,64) 
 
*2.Trabajo y seguridad social* 
gen  miss_dim2_ind1=(p20==. & p21==. & p22==.) if inrange(p03,5,14) 
replace miss_dim2_ind1=1 if condactn==. & inrange(p03,15,17) 
replace miss_dim2_ind1=1 if condactn==1 & inrange(p03,15,17) & (p07==. | horas==.) 
egen    miss_dim2_ind2=rowmiss(condactn) if inrange(p03,18,98) 
replace miss_dim2_ind2=1 if p20==. & p21==. & p22==. & p32==. & p34==. & p35==. & /// 
inrange(p03,18,98) 
egen    miss_dim2_ind3=rowmiss(p72a p75 p77 p05a p05b) if inrange(p03,15,98) 
 
*3.Salud, agua y alimentación* 
egen miss_dim3_ind1=rowmiss(epobre) 
egen miss_dim3_ind2=rowmiss(vi10) 
 
*4.Hábitat, vivienda y ambiente sano* 
egen miss_dim4_ind1=rowmiss(hsize vi07) 
egen miss_dim4_ind2=rowmiss(vi03a vi03b vi04a vi04b vi05a vi05b) 
egen miss_dim4_ind3=rowmiss(vi09) 
egen miss_dim4_ind4=rowmiss(vi13) 
 
egen miss_total=rowtotal(miss_dim*) 
recode miss_total (1/max=1) 
label var miss_total "Total de hogares sin información" 
 
drop miss_dim* 
drop if p03==99 
tempvar aux 
bys id_hogar: egen `aux'=max(miss_total) 
replace miss_total=`aux' 
drop if miss_total==1 
 
*=================================================================================* 
* C. ESTRUCTURA DE PONDERACIÓN 
*=================================================================================* 
 
# delimit ; 

mat pesos=(1/4*1/3,1/4*1/3,1/4*1/3, 
     1/4*1/3,1/4*1/3,1/4*1/3, 
     1/4*1/2,1/4*1/2, 
     1/4*1/4,1/4*1/4,1/4*1/4,1/4*1/4) ; 
# delimit cr;      
svmat double pesos, names(wdim) 
foreach var of varlist wdim* { 
replace `var'=`var'[_n-1] if _n>1 
} 
 
local n2=1 
foreach var of varlist dim*_h { 
gen wp_`var'=`var'*wdim`n2++' 
} 
 
egen ci=rsum(wp_*) 
label var ci "Vector de conteo" 
*=================================================================================* 
* D. LINEAS DE POBREZA MULTIDIMENSIONAL Y COMPONENTES DEL IPM 
*=================================================================================* 
gen   TPM  =(ci>=4/12) 
gen   TPEM =(ci>=6/12) 
gen   A   =ci   if TPM==1 
gen   IPM =0  
replace  IPM =ci   if TPM==1  
 
label var  TPM  "Tasa de Pobreza Multidimensional (k>=1/3)" 
label var  TPEM "Tasa de Pobreza Extrema Multidimensional (k>=1/2)" 
label var  A    "Porcentaje Promedio de Privaciones (k>=1/3)" 
label var  IPM  "Índice de Pobreza Multidimensional (k>=1/3)" 
 
keep TPM TPEM A IPM id_per 
save `bddf', replace 
 
use `bdd1', clear 
merge 1:1 id_per using `bdd3', keepusing(dim*_p) keep(1 3) nogen 
merge 1:1 id_per using `bddf', keep(1 3) nogen 
order id_*, first 
 
*=================================================================================* 
*    FIN 
*=================================================================================* 

 


